# Maintainer: Dave Reisner <dreisner@archlinux.org>
# Contributor: Stéphane Gaudreault <stephane@archlinux.us>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>
# Contributor: damir <damir@archlinux.org>

pkgname=time
pkgver=1.9
pkgrel=4
pkgdesc="Utility for monitoring a program's use of system resources"
arch=('loong64' 'x86_64')
url='https://www.gnu.org/software/time/'
license=('GPL')
depends=('glibc')
source=(https://ftp.gnu.org/gnu/time/time-$pkgver.tar.gz{,.sig})
sha256sums=('fbacf0c81e62429df3e33bda4cee38756604f18e01d977338e23306a3e3b521e'
            'SKIP')
validpgpkeys=('F576AAAC1B0FF849792D8CB129A794FD2272BC86')

build() {
  cd "$pkgname-$pkgver"

  ./configure --prefix=/usr --infodir=/usr/share/info
  make
}

check() {
  make -C "$pkgname-$pkgver" check
}

package() {
  cd "$pkgname-$pkgver"

  make exec_prefix="$pkgdir/usr" infodir="$pkgdir/usr/share/info" install
}
